

# Perl III

File I/O, more on system calls

Dave Messina

v4 post 2012

1

# File I/O

I/O stands for input/output.

It's how get computer programs talk  
to the rest of the world.

2

# Perl has magic

Perl has a magic way that makes it super easy to get data from files and into your program.

It looks like this: <>

3



<> will:

read filenames that are arguments on the command line

open each file in turn

read each line from the file

4



```
#!/usr/bin/perl
# how to read a file with <>
use warnings;
use strict;

while (my $line = <>) {
    chomp $line;
    print "Here's a line: ", $line, "\n";
}
```

5

## Sidebar: chomp

chomp removes the newline from the end of a string (if there is a newline).

```
my $string = "hey there!\n";
print "my string is: ", $string, "\n";
chomp $string;
print "after chomp : ", $string, "\n";
```

When you read a file, the first thing you always want to do is chomp.

6



Let's make a file and read from it.  
We'll call it myfile.txt

```
% perl read_from_file.pl myfile.txt
```

And now we're giving the name myfile.txt as a  
command-line argument to our Perl script.

7

## <> line count

Let's do something more interesting than  
printing the line back out. Let's count how  
many lines there are in the file.

```
my $line_count;  
while (my $line = <>) {  
    chomp $line;  
    $line_count++;  
}  
print "There are $line_count lines\n";
```

8

## Sidebar: increment operators

Yesterday we learned several numeric operators.  
Here are a couple more common ones:

**++** the increment operator

```
my $x = 1;  
$x++;      # add 1 to $x
```

```
# exactly the same as  
$x = $x + 1;
```

9

## Sidebar: decrement operators

**--** the decrement operator

```
my $x = 1;  
$x--;      # subtract 1 from $x
```

```
# exactly the same as  
$x = $x - 1;
```

10

## <> line count

With ++, we're counting each time we go through the loop.

```
my $line_count;
while (my $line = <>) {
    chomp $line;
    $line_count++;
}
print "There are $line_count lines\n";
```

11

## <> multiple files

If there is more than one argument, each one is opened and read completely, one after the other.

```
% perl read_from_file.pl myfile.txt another.txt
```

So let's create another file and try it.

12

## <> mistakes

Remember how yesterday we had command-line arguments that were numbers?

Does Perl know that the arguments are files?

```
% perl read_from_file.pl 2 9
```

Let's try it and see what happens.

13

## the input loop

Let's step back for a moment and think about why <> works. What is while? What is it testing?

```
my $line_count;  
while (my $line = <>) {  
    chomp $line;  
    $line_count++;  
}  
print "There are $line_count lines\n";
```

14

# the input loop

What exactly is going on on this line?

```
while (my $line = <>) {
```

The `<>` is a function.

It returns a line of input.

We assign that line to a variable, `$line`.

While tests that assignment for truth:

"Can we assign a value to `$line`?"

15

# the input loop

If there is another line in the file, the answer is "yes, we can, it's TRUE."

If we've hit the end of the file, there are no more lines to read, and so the answer is "no", or FALSE.

When the expression in parentheses is false, we exit the loop.

16



# the input loop

Once we've exited the loop, the print statement gets executed.

```
my $line_count;
while (my $line = <>) {
    chomp $line;
    $line_count++;
}
print "There are $line_count lines\n";
```

17

# the input loop

To summarize:

The while loop will read one line of text after another. At the end of input, the <> operator returns undef and the while loop terminates.

Remember that even blank lines in a file are TRUE, because they consist of a single newline character.

18

# STDOUT and STDERR

Every Perl script by default has two places it knows where to write to:

STDOUT and STDERR

19

# STDOUT and STDERR

## STDOUT

Standard output, used to write data out.

Prints to your screen, but can be redirected to a file or other program from the shell using redirection or pipes.

20

# STDOUT and STDERR

## STDERR

Standard error, used for diagnostic messages.

Also prints to your screen, and also can be redirected to a file or other program from the shell using redirection or pipes.

21

# STDOUT and STDERR

You've actually been usually STDOUT all along. It's the default place where your program's output goes.

When you use `print`, you're actually writing to STDOUT.

These are equivalent:

```
print "Well, how did I get here?\n";  
print STDOUT "Well, how did I get here?\n";
```

22

# STDOUT and STDERR

But you can also specify other places to write to.

Like STDERR:

```
print STDOUT "You may ask yourself:\n";  
print STDERR "Well, how did I get here?\n";
```

23

# STDOUT and STDERR

At first it looks exactly the same as STDOUT, but if we use output redirection on the command line, we can see that the output is actually going to a different place:

```
$ perl test.pl > output.txt  
Well, how did I get here?
```

24

# open for reading

<> is great, but often you want to read from a specific file. You can do that using open.

```
my $file = shift;  
open(my $filehandle, '<', $file) or die "can't open  
$file: $!\n";
```

25

# open

```
my $file = shift @ARGV;  
open(my $filehandle, '<', $file)  
    or die "can't open $file: $!\n";
```

**Let's break this down into pieces:**

```
my $file = shift @ARGV;
```

**reads the filename from the command line.**

26

# open

```
open(my $filehandle, '<', $file)
```

open is a function, which takes 3 arguments:

The first argument is a filehandle. Filehandles are how you refer to a file within Perl. STDOUT and STDERR are filehandles.

When you open a file yourself, you make your own filehandle and give it a name (here, I chose \$filehandle).

27

# open

```
open(my $filehandle, '<', $file)
```

The second argument is a mode. The modes are borrowed from redirection on the command line.

- < for reading from a file
- > for writing to a file

28

# open

```
open(my $filehandle, '<', $file)
```

The third argument is the name of a file to open. It can either be a literal name:

```
open(my $filehandle, '<', 'myfile.txt')
```

or a variable containing a filename:

```
open(my $filehandle, '<', $file)
```

Where can you go for more information on open?

29

# open or die

```
or die "can't open $file: $!\n";
```

open or die is a Perl idiom. die is a function that exits the program immediately and prints the specified string to STDERR.

Why or? What is being tested for truth?

30

# open — \$!

```
or die "can't open $file: $!\n";
```

\$! is a special Perl variable that contains error messages from the system. If there was a problem with opening your file, there will be an error message in \$!, and we can include it in *our* error string.

Let's try it.

31

# open for writing

Open also can be used to open files for *writing* by using '>' as the second argument to open.

```
my $out = shift @ARGV;  
open(my $filehandle, '>', $out)  
  or die "can't open $out: $!\n";
```

Now specify that filehandle when you print:

```
print $filehandle "I'm writing to a file!\n";
```

**Be careful! If you open an existing file for writing, you will erase everything inside that file!**

32



# open

You can open more than one file in a script — just give them different filehandles.

```
my $in = shift @ARGV;
my $out = 'out.txt';
open(my $in_fh, '<', $in) or die "can't open $in: $!\n";
open(my $out_fh, '>', $out) or die "can't open $out: $!\n";
```

33

# open

To read from a filehandle line by line, you put the name of the filehandle inside <>, like this:

```
my $in = shift @ARGV;
open(my $in_fh, '<', $in) or die "can't open $in: $!\n";

while (my $line = <$in_fh>) {
    chomp $line;
    print "This line is from the file $in: $line\n";
}
```

34

# a quick word on system

We saw yesterday that there were two ways of executing a command line from within Perl:

```
# with system
system("sort $file");

# or with backticks
`sort $file`;
```

35

# a quick word on system

With backticks, you can capture the output from the command into a variable:

```
open(my $out_fh, '>', 'sorted.txt')
    or die "error:$!";
my $sorted_output = `sort $file`;
print $out_fh "sorted output:\n", $sorted_output;
```

36